```
STUDENT > # Solve the system using mechanical parameters and using
          electrical analogy
STUDENT >
STUDENT > # Contents
STUDENT > # Part 1 = Symbols
STUDENT > # Part 1a = Initialization
STUDENT > # Part 2 = Solve the system mechanically to determine
          Hmech(s)
STUDENT > # Part 3 = Solve the system using electrical analogy to
          determine Helec(s)
STUDENT > # Part 3a -  verify Hmech(s) = Helec(s)
STUDENT > # Part 3b - determine roots from electrical solution
STUDENT > # Part 3c - analyse max power transfer using electrical
          solution
STUDENT > # Part 4 - Conclusion - use Rmax
STUDENT > # Part 5 - Compare part 4 strategy to adjustment of R to
          obtain rsonance
STUDENT >
STUDENT > #===== PART 1 - SYMBOLS==========
STUDENT > # capital letters R, L, C for electrical parameters
STUDENT > # small letters m, k, c for mechanical parameters
STUDENT >
STUDENT >
STUDENT > # Mechanical solution variables
STUDENT > # X(s) = displacement of the mass
STUDENT > # Xwave(s) = displacement of the wave
STUDENT > # Hmech(s) = X/Xwave  - transfer functio
STUDENT >
STUDENT > # Electrical solution variables
STUDENT > # Vmid(s) = voltage at midpoint node between L and R/C
STUDENT > # Vin(s) = voltage of the wave
STUDENT > # Helec(s) = Vmid / Vin - transfer function
STUDENT > # P = power
STUDENT > # Rmax = R chosen to maximize power
STUDENT > # dPdR = dP/dR
STUDENT > # dP2dR2 = d^2P/dR^2
STUDENT >
STUDENT > # ====Part 1a === Initialize=====
STUDENT > restart;
STUDENT > # Correspondence between the mech and elec parameters is
          given in the following, in a forma suitable for use with
          the subs substitution function:
STUDENT > subsM2E:={k=1/L,c=1/R,m=C};
```

$$subsM2E := \{m = C, c = \frac{1}{R}, k = \frac{1}{L}\}$$

```
STUDENT > subsE2M:={L=1/k,R=1/c,C=m};
```

$$subsE2M := \{C = m, R = \frac{1}{c}, L = \frac{1}{k}\}$$

```
STUDENT > #====Part 2 -Solve the mechanical system ================
STUDENT > # Sum of forces acting upward on the node at location of
          mass is acceleration
STUDENT > eq1:=(Xwave-X)*k  -c*s*X = m*s^2*X;
```

$$eq1 := (Xwave - X)\,k - c\,s\,X = m\,s^2\,X$$

```
STUDENT > eq1:=simplify(lhs(eq1)-k*Xwave-m*s^2*X =
          rhs(eq1)-k*Xwave-m*s^2*X);
```

$$eq1 := -k\,X - c\,s\,X - m\,s^2\,X = -k\,Xwave$$

```
STUDENT > eq1:=simplify(-1*eq1);
```

$$eq1 := k\,X + c\,s\,X + m\,s^2\,X = k\,Xwave$$

```
STUDENT > # Let H(s) = X/Xwave
STUDENT > H(s):=k/(m*s^2+c*s+k);
```

$$\mathrm{H}(s) := \frac{k}{m\,s^2 + c\,s + k}$$

```
STUDENT > # ========Part 3 - Solve Electrical
          Analogy================
STUDENT > # Now solve the same system using electrical analogy:
STUDENT > ZC:=1/(s*C);
```

$$ZC := \frac{1}{s\,C}$$

```
STUDENT > ZR:=R;
```

$$ZR := R$$

```
STUDENT > ZL:=s*L;
```

$$ZL := s\,L$$

```
STUDENT > Zpara:=ZC*ZR/(ZC+ZR);
```

$$Zpara := \frac{R}{s\,C\left(\frac{1}{s\,C} + R\right)}$$

```
STUDENT > Zpara:=simplify(Zpara);
```

$$Zpara := \frac{R}{1 + R\,s\,C}$$

```
STUDENT > Vmid:=Vin*Zpara/(ZL+Zpara); # voltage at the midpoint of
          the ckt between L and R/C
```

$$Vmid := \frac{Vin\, R}{(1 + R\, s\, C)\left(s\, L + \frac{R}{1 + R\, s\, C}\right)}$$

```
STUDENT > Vmid:=simplify(Vmid);
```

$$Vmid := \frac{R\, Vin}{s\, L + s^2\, L\, R\, C + R}$$

```
STUDENT > Helec:=Vmid/Vin;
```

$$Helec := \frac{R}{s\, L + s^2\, L\, R\, C + R}$$

```
STUDENT > #================ Part 3A - Compare Hmech to Helec=========
STUDENT > # Compare to mechanical system by susbsituting in the equivalence defined above subsE2M
STUDENT > simplify(subs(subsE2M,Helec));
```

$$\frac{k}{m\, s^2 + c\, s + k}$$

```
STUDENT > # This matches Hmech(s) as expected
STUDENT > # Therefore we get the SAME transfer function when we solve using electrical analogy as when we solved using mechanical parameters directly.  (actually mechanical transfer function was ratio of displacements where electrical voltage ratio corresponded to ratio of velocities, but ratio of velocities is same as ratio of displacements)
STUDENT >
STUDENT > # ====Part 3B-Find Poles=Roots ==================
STUDENT > TheElecRoots:=solve(denom(Vmid)=0,s);
```

$$TheElecRoots := \frac{1}{2}\,\frac{-L + \sqrt{L^2 - 4\, L\, R^2\, C}}{L\, R\, C},\ \frac{1}{2}\,\frac{-L - \sqrt{L^2 - 4\, L\, R^2\, C}}{L\, R\, C}$$

```
STUDENT > wde:=simplify(TheElecRoots[1]+1/(2*R*C));  # wd from electrical analogy
```

$$wde := \frac{1}{2}\,\frac{\sqrt{L\,(L - 4\, R^2\, C)}}{L\, R\, C}$$

```
STUDENT > wdm:=simplify(subs(subsE2M,wde));
```

$$wdm := \frac{1}{2}\,\frac{\sqrt{\frac{c^2 - 4\, m\, k}{k^2\, c^2}}\; k\, c}{m}$$

```
STUDENT >
STUDENT > #======Part 3c-Analyse Max Power Transfer=========
STUDENT > # The objective is to maximixe the Power
STUDENT > # Subsitute s=I*w for sinusoidal waveform
STUDENT > Vmid:=subs(s=I*w,Vmid);
```

$$Vmid := \frac{R\,Vin}{I\,w\,L - w^2\,L\,R\,C + R}$$

```
STUDENT > Vmidmag:=simplify(evalc(abs(Vmid)));  # magnitude of Vmid
```

$$Vmidmag := \sqrt{\frac{R^2\,Vin^2}{w^4\,L^2\,R^2\,C^2 - 2\,w^2\,L\,R^2\,C + R^2 + w^2\,L^2}}$$

```
STUDENT > P:=(1/2)*Vmidmag^2/R;  #power
```

$$P := \frac{1}{2}\,\frac{R\,Vin^2}{w^4\,L^2\,R^2\,C^2 - 2\,w^2\,L\,R^2\,C + R^2 + w^2\,L^2}$$

```
STUDENT > dPdR:=diff(P,R);
```

$$dPdR := \frac{1}{2}\,\frac{Vin^2}{w^4\,L^2\,R^2\,C^2 - 2\,w^2\,L\,R^2\,C + R^2 + w^2\,L^2} - \frac{1}{2}\,\frac{R\,Vin^2\,(2\,w^4\,L^2\,R\,C^2 - 4\,w^2\,L\,R\,C + 2\,R)}{(w^4\,L^2\,R^2\,C^2 - 2\,w^2\,L\,R^2\,C + R^2 + w^2\,L^2)^2}$$

```
STUDENT > Rmax:=solve(dPdR=0,R);
```

$$Rmax := -\frac{w\,L}{w^2\,L\,C - 1},\ \frac{w\,L}{w^2\,L\,C - 1}$$

```
STUDENT > # Which one is positive depends on which side of the
          undamed natural freq 1/sqrt(LC) is w
STUDENT >
STUDENT > Rmax:=Rmax[1];  # choose the root which is positive
          assuming w^2LC-1 > 0 (above undamed nat freq)
```

$$Rmax := -\frac{w\,L}{w^2\,L\,C - 1}$$

```
STUDENT > dP2dR2:=diff(dPdR,R);
```

$$dP2dR2 := -\frac{Vin^2\,(2\,w^4\,L^2\,R\,C^2 - 4\,w^2\,L\,R\,C + 2\,R)}{\%1^2} + \frac{R\,Vin^2\,(2\,w^4\,L^2\,R\,C^2 - 4\,w^2\,L\,R\,C + 2\,R)^2}{\%1^3} - \frac{1}{2}\,\frac{R\,Vin^2\,(2\,w^4\,L^2\,C^2 - 4\,w^2\,L\,C + 2)}{\%1^2}$$

$$\%1 := w^4\,L^2\,R^2\,C^2 - 2\,w^2\,L\,R^2\,C + R^2 + w^2\,L^2$$

```
STUDENT > simplify(subs(R=Rmax,dP2dR2));
```

$$\frac{1}{4}\,\frac{(w^2\,L\,C - 1)\,Vin^2}{L^3\,w^3}$$

```
STUDENT > # The 2nd derivative is negative (under same assumption
          w^2LC-1 is + as above), so this is a relative maximum of
```

```
power
STUDENT > cmax:=simplify(subs(subsE2M,Rmax));
```

$$cmax := -\frac{w}{w^2\, m - k}$$

```
STUDENT >
STUDENT > # ============Part 4 = Conclusion=========
STUDENT > # Conclusion - maximize power by sleecting
          R=Rmax=w*L/(w^2*L*C-1)
STUDENT > #  equivalently selecting c = cmax = w/(w^2*m-k)
STUDENT >
STUDENT >
STUDENT > #=== Part 5 - Compare setting R to Rmax vs setting R to
          achieve resonance - see if they are the same ====
STUDENT > # If w is excitatio frequency, part 4 tells us to set
          R=Rmax(w)
STUDENT >
STUDENT > # Find the R to make wde = w and see if it is equal to
          Rmax
STUDENT > Rresonant:=solve(wde=w,R);
```

$$Rresonant := 2\,\frac{\sqrt{C\,(w^2\,L\,C+1)\,L}}{4\,C+4\,w^2\,L\,C^2},\ -2\,\frac{\sqrt{C\,(w^2\,L\,C+1)\,L}}{4\,C+4\,w^2\,L\,C^2}$$

```
STUDENT > simplify(Rresonant[1]);
```

$$\frac{1}{2}\,\frac{\sqrt{C\,(w^2\,L\,C+1)\,L}}{C\,(w^2\,L\,C+1)}$$

```
STUDENT > Compare to Rmax
STUDENT > Rmax;
```

$$-\frac{w\,L}{w^2\,L\,C-1}$$

```
STUDENT > # Rresonant does not appear to be the same as Rmax .
STUDENT > # Recommend using the Rmax formulation of part 4, rather
          than adjusting R to resonance
```